
ABSTRACT 

Research studies often involve continuous outcomes, for example survey instrument scores measured 
repeatedly over time. We see this in our cancer research when patients complete surveys before, during, 
and after an intervention. Our motivating example evaluates an intervention vs control, within a 
randomized trial, and compares scores between groups at 3 timepoints. PROC GENMOD is used to fit 
marginal linear models with generalized estimating equation methods to account for potential clustering 
within patients and allow for missingness. The longitudinal model includes time, group (intervention v 
control), and the interaction between time and group. There are many interesting estimates from this type 
of study, including 1) outcome scores at each time point, 2) differences in scores between groups at 
different timepoints, 3) changes in scores over time within groups, and 4) differences in the slopes of 
scores over time by group. Additionally, consideration must be made in treating time as continuous or 
categorical.  PROC GENMOD is a useful tool to make these comparisons, but special considerations 
must be made in order to write all the necessary ESTIMATE statements. While tools like LSMEANS are 
useful, understanding how to write ESTIMATE statements for comparisons is helpful for cleaning output 
and making comparisons that are unable to be made via LSMEANS. We lay out the underlying model 
structure and equations to clearly illustrate how to write these ESTIMATE statements.  

1 INTRODUCTION 

Many studies use a continuous outcome measure, for example a survey instrument with continuous 
scores, and repeat the measure over time. The goal of these studies is generally to examine how an 
intervention might change a continuous outcome (patient self-efficacy, anxiety, depression, lab values, 
etc.) over time, separately for intervention and control groups. Additionally, studies aim to examine if the 
outcomes change differently over time and if there are group differences at various timepoints. Our 
motivating example is a pilot randomized control trial which evaluated the Families Addressing Cancer 
Together (FACT) intervention1. FACT is a web‐based, individually tailored, psychoeducational 

intervention for parents with cancer to improve illness‐related communication with their minor children. 
The outcome measure was the Communication Self-Efficacy Scale (CSES), which had scores ranging 
from 0-100, and was asked at baseline (t0; 0 weeks), mid-intervention (t1; 3 weeks), and post-intervention 
(t2; 6 weeks). Higher scores reflected more confidence in communication self-efficacy. 

For analyzing longitudinal data with repeated measures, generalized estimating equation (GEE) methods 
developed by Liang and Zeger, is a common method for appropriately handling correlated outcome data 
collected over time2. This method requires specification of a working correlation matrix, a link function, 
and a distributional assumption. Common working correlation matrices are independent, exchangeable, 
autoregressive, and unstructured and can be chosen based on what assumptions can be made about the 
correlations. Many link functions may also be specified, such as logistic, linear, log-linear, etc. depending 
on the type of outcome data. The focus for this is a marginal linear model with GEE methods, using a 
linear link function with a normal distributional assumption and an exchangeable correlation matrix. PROC 
GENMOD is a common SAS procedure that is utilized to build this model type3. A common issue with 
PROC GENMOD for investigators is the mystification of ESTIMATE statements. We aim to add to the 
existing literature with specific examples that add explanation about the underlying model to develop 
ESTIMATE statements to produce results of interest in clinical practice. While PROC GENMOD has many 
useful statements, such as LSMEANS, that are able to provide differences in least squares means, it is 
important to understand how to build ESTIMATE statements for ease of interpretation, clean output 
production, and estimating comparisons that aren’t available via LSMEANS.  



2 DATA STRUCTURE 

The data structure for the motivating example is a long data form. There is a row for each repeated 
measure containing the patient ID (index), the group (FACT intervention vs. control), the timepoint 
(baseline = t0, mid-intervention = t1, and post-intervention = t2), and their CSES score.  

See the sample data structure below:  

Index Group Timepoint (t) CSES 

1 Intervention (FACT) 0 95.5 

1 Intervention (FACT) 1 97.8 

1 Intervention (FACT) 2 100 

4 Control 0 69.4 

4 Control 1 66.7 

4 Control 2 76.1 

 

2.1 MODEL TYPE AND TIME SPECIFICATION 

Marginal models were used with generalized estimating equation methods to account for potential 
clustering within patients. A normal distribution with an identity link function was used to build the 
model. All longitudinal models included the following covariates: time (baseline (t0), mid-intervention 
(t1; 3 weeks), post-intervention (t2; 6 weeks)), the grouping variable (FACT intervention vs. control), 
and the interaction between time and the grouping variable. The time variable can be treated as 
continuous or categorical. It may be beneficial to treat time categorically if the outcome is not 
monotonically increasing or decreasing over time. The first step should be to graph the raw data and 
connect lines over time for each patient. If the graph suggests, for example, that scores dip or rise at t1 
compared to t0 and t2, then treating time as categorical is likely more appropriate. If the graphs show 
fairly straight lines for each patient, treating time as continuous is likely appropriate. 

3 CONTINUOUS TIME 

When treating time as a continuous variable, the model can be written as:  

𝑔(𝜇𝑖𝑡) = 𝑎 + 𝛽1𝑔𝑟𝑜𝑢𝑝 +  𝛽2𝑡𝑖𝑚𝑒 +  𝛽3𝑔𝑟𝑜𝑢𝑝 ∗ 𝑡𝑖𝑚𝑒 
 
Where g(.) is the specified link function, 𝜇𝑖𝑡 is the mean CSES score at time t for subject i (a continuous 
score representing communication self-efficacy), α is the intercept, β1 is the parameter estimate for the 
grouping variable which represents the difference in the FACT intervention vs. control at time 0, β2 is the 
parameter estimate for the time variable representing the slope of the control group from time 0 to time 1, 
as well as time 1 to time 2, β3 is the parameter estimate for the interaction term representing the 
difference in slope between the two study groups. This model framework can be used to check 
ESTIMATE statements and identify the needed parameters.  

The following SAS code represents the model where index denotes the subject’s identifier, and the 
REPEATED statement allows for GEE use. Reference coding was used for this study.  

proc genmod data=dataset; 

class group(ref="Control (waitlist)") index;     

 model outcome = group t group*t/ dist=normal link=identity; 

 repeated subject=index / type=exch; 

run; 

 



The general form of the ESTIMATE statements is as follows, where terms are included and excluded 
based on comparisons of interest:  

ESTIMATE "Description" int group t group*t; 

When running the code above, the following output is produced detailing the parameter information and 
the parameter estimates (Figure 1 and Figure 2). Using the equation derived in the next section, the 
parameter estimates can be used to check the ESTIMATE statement results.  

Figure 1. Parameter information detailing the order of the parameters in the model. 

 

Figure 2. Parameter estimates after running the PROC GENMOD procedure. 

 

3.1 COMPARISONS OF INTEREST 

For a study with two groups and multiple timepoints, there are multiple comparisons one might be 
interested in. These are the typical research questions our investigators often want to answer: 

1. What are the model estimated outcome measure scores at each time point for each group? 

2. Are the outcome measure scores different by group at any timepoint?  

3. Is there a significant change over time in outcome measure within group (separately for 
control and FACT intervention)? 

4. Does the outcome measure change differently over time by group? (i.e. is there a difference 
in slopes between the FACT intervention and control groups)? 

For each of these comparisons, we will outline the appropriate equations that can be written and the 
accompanying SAS code needed to form the corresponding ESTIMATE statements. 

  



3.1.1 ESTIMATED SCORES AT EACH TIMEPOINT 

CONTROL: Model estimates for control at time =0, time=1, and time=2 

• Estimate for control at time=0; Let group = 0, time = 0, so group*time=0 
= 𝑎 + 𝛽10 +  𝛽20 +  𝛽30 ∗ 0 

= 𝑎 

= 79.5 

• Estimate for control at time=1; Let group = 0, time = 1, so group*time=0 
= 𝑎 + 𝛽10 +  𝛽21 +  𝛽30 ∗ 0 

= 𝑎 + 𝛽2 

= 79.5 + 0.3 

= 79.8 

• Estimate for control at time=2; Let group = 0, time = 2, so group*time=0 
= 𝑎 + 𝛽10 +  𝛽22 +  𝛽30 ∗ 0 

= 𝑎 + 2𝛽2 

= 79.5 + 2(0.3) 

= 80.1 

The necessary ESTIMATE statements are: 
ESTIMATE "model est control at t0" int 1 group 0 1 t 0 group*t 0 0; 

ESTIMATE "model est control at t1" int 1 group 0 1 t 1 group*t 0 1; 

ESTIMATE "model est control at t2" int 1 group 0 1 t 2 group*t 0 2;  

 

To write the statement for control at time 0, we include the intercept, time, group, and the group*time 

interaction term in the ESTIMATE statements. The intercept is included as ‘int 1’, derived directly from 

the linear combinations in the equations above. In Figure 1, the parameter information is ordered, 

where the control group (reference) is listed second as Prm3 in comparison to Prm2 representing the 

FACT intervention group. Therefore, to indicate the control group, we use the second position of the 

group section, giving ‘group 0 1.’ Using continuous time, time will take the value of 0, resulting in ‘t 0’ 

as the required syntax. For the interaction term, insert the continuous time value into the second 

position to achieve ‘group*t 0 0’. The ESTIMATE statements for the control group at time 1 and 2 

follow the same logic. When checking, we see that the ESTIMATE output in Figure 3 matches the 

calculations from the formulation above using the parameter estimates in Figure 2. 

 

INTERVENTION: Model estimates for FACT intervention at time =0, time=1, and time=2 

• Estimate for FACT at time=0; Let group = 1, time = 0, so group*time=0 
= 𝑎 + 𝛽11 + 𝛽20 + 𝛽30 ∗ 0 

= 𝑎 + 𝛽1 

= 79.5 + 1.2 

= 80.7  

• Estimate for FACT at time=1; Let group = 1, time = 1, so group*time=1 
= 𝑎 + 𝛽11 + 𝛽21 + 𝛽31 ∗ 1 

= 𝑎 + 𝛽1 + 𝛽2 + 𝛽3 

= 79.5 + 1.2 + 0.3 + 4.4 

= 85.4 

• Estimate for FACT at time=2; Let group = 1, time = 2, so group*time=2 
= 𝑎 + 𝛽11 + 𝛽22 + 𝛽31 ∗ 2 

= 𝑎 + 𝛽1 + 2𝛽2 + 2𝛽3 

= 79.5 + 1.2 + 2(0.3) + 2(4.4) 

= 90.1 

The necessary ESTIMATE statements are as follows:  
ESTIMATE "model est FACT at t0" int 1 group 1 0 t 0 group*t 0 0; 

ESTIMATE "model est FACT at t1" int 1 group 1 0 t 1 group*t 1 0;  



ESTIMATE "model est FACT at t2" int 1 group 1 0 t 2 group*t 2 0;  

 

To write the statement for the FACT intervention at time 0, we include the intercept, time, group, and 

the group*time interaction term in the ESTIMATE statements. The intercept is included as ‘int 1’, 

derived directly from the linear combinations in the equations above. In Figure 1, the parameter 

information is ordered, where the FACT intervention group is listed first as Prm2, in comparison to 

Prm3 representing the control group (reference). Therefore, to indicate the FACT intervention group, 

we use the first position of the group section, giving ‘group 1 0.’ Using continuous time, time will take 

the value of 0, resulting in ‘t 0’ as the required syntax. For the interaction term, insert the continuous 

time value into the first position to achieve ‘group*t 0 0’. The ESTIMATE statements for the 

intervention group at time 1 and 2 follow the same logic. When checking, we see that the ESTIMATE 

output in Figure 3 matches the calculations from the formulation above using the parameter estimates 

in Figure 2. 

3.1.2 DIFFERENCES IN SCORES AT EACH TIMEPOINT 

Are the outcome measure scores different by group at any time (for example, at time 2)?  

• Let group = 1, time = 2, so group*time=2 
= 𝑎 + 𝛽11 +  𝛽22 +  𝛽31 ∗ 2 

= 𝑎 + 𝛽1 + 2𝛽2 + 2𝛽3 

= 79.5 + 1.2 + 2(0.3) + 2(4.4) 

• Let group = 0, time = 2, so group*time=0 
= 𝑎 + 𝛽10 +  𝛽22 +  𝛽30 ∗ 0 

= 𝑎 + 2𝛽2 

= 79.5 + 2(0.3) 

 

To find the difference between these two groups we subtract:  
= (𝑎 + 𝛽1 + 2𝛽2 + 2𝛽3) − (𝑎 + 2𝛽2) 

= 𝛽1 + 2𝛽3 

= 1.2 + 2(4.4) 

= 10.0 

𝛽1 + 2𝛽3 represents the difference between FACT at time=2 and control at time=2 

 

The necessary ESTIMATE statement is: 
ESTIMATE "Diff FACT vs. control at t2" group 1 -1 group*t 2 -2;  

 

To write the statement for the difference in average CSES between the FACT intervention and 

control groups at time 2, we include the group, and the group*time interaction term in the ESTIMATE 

statements. Note that 𝛽1 and 𝛽3 correspond to these terms in the model. The term for the intercept 

and time are not needed because they drop out of the model. In Figure 1, the parameter information 

is ordered, where the FACT intervention group is listed first as Prm2, in comparison to Prm3 

representing the control group (reference). In SAS, when a 1:1 comparison is made between two 

groups, the syntax is 1 for the main group and -1 for the comparator group, thus ‘group 1 -1’ is 

utilized to represent the comparison between the FACT intervention group and control group. 

Similarly, the interaction term follows the same syntax indicating a comparison, but time is multiplied 

through the interaction term which gives ‘group*t 2 -2’. When checking, we see that the ESTIMATE 

output in Figure 3 matches the calculations from the formulation above using the parameter 

estimates in Figure 2. 

 

Similarly, the ESTIMATE statements for time 0 and time 1 are achieved:  
ESTIMATE "Diff FACT vs. control at t0" group 1 -1; 

ESTIMATE "Diff FACT vs. control at t1" group 1 -1 group*t 1 -1; 



3.1.3 CHANGE OVER TIME WITHIN GROUP 

Since we are treating time as continuous, the model is fitting a line with a single slope for each 
group. 

CONTROL: For the control group, group=0. We want to compare the change from t0 to t1.   

• Let group = 0 and time = 1, so group*time=0  
= 𝑎 + 𝛽10 + 𝛽21 +  𝛽30 ∗ 1 = 𝑎 + 𝛽2 

=  79.5 + 0.3 

• Let group = 0 and time = 0, so group*time=0 
= 𝑎 + 𝛽10 +  𝛽20 +  𝛽30 ∗ 0 

= 𝑎 

= 79.5 

To find the difference between these two groups we subtract:  
= (𝑎 + 𝛽2) − (𝑎) = 𝛽2 

= 0.3 

𝛽2 represents the difference between control at time=1 and control at time=0. (i.e. slope for control) 
The necessary ESTIMATE statement is: 
ESTIMATE "slope for control" t 1 group*t 0 1; 

 

To write the statement for the change over time in average CSES in the control group between time 

0 and time 1, we include the time and the group*time interaction term in the ESTIMATE statements. 

In Figure 1, the parameter information is ordered, where the FACT intervention group is listed first as 

Prm2, in comparison to Prm3 representing the control group (reference). Using continuous time, 

time will take the value of 1, resulting in ‘t 1’ as the required syntax since we are comparing time 1 to 

time 0. For the interaction term, insert the continuous time value into the second position to achieve 

‘group*t 0 1’. The ESTIMATE statement value for the slope of the control group from time 0 and time 

1 will be equivalent to that of the slope of the control group from time 1 to time 2, given the linear 

nature of the model with a constant rate of change. When checking, we see that the ESTIMATE 

output in Figure 3 matches the calculations from the formulation above using the parameter 

estimates in Figure 2. 

 

INTERVENTION: For the FACT intervention group, group=1, and we want to compare the change 

from t0 to t1.  

 

• Let group = 1, time = 1, so group*time=1  
= 𝑎 + 𝛽11 +  𝛽21 +  𝛽31 ∗ 1 

= 𝑎 + 𝛽1 + 𝛽2 + 𝛽3 

= 79.5 + 1.2 + 0.3 + 4.4 

• Let group = 1, time = 0, so group*time=0 
= 𝑎 + 𝛽11 +  𝛽20 +  𝛽30 ∗ 0 

= 𝑎 + 𝛽1 

= 79.5 + 1.22 

To find the difference between these two groups we subtract:  
= (𝑎 + 𝛽1 + 𝛽2 + 𝛽3) − (𝑎 + 𝛽1) 

= 𝛽2 +  𝛽3 

= 0.3 + 4.4 

= 4.7 

𝛽2 +  𝛽3 represents the difference between FACT at time=1 and FACT at time=0. (i.e. slope for 
FACT)  
The necessary ESTIMATE statement is: 
ESTIMATE "slope for FACT" t 1 group*t 1 0; 



To write the statement for the change over time in average CSES in the FACT intervention group 

between time 0 and time 1, we include the time and the group*time interaction term in the 

ESTIMATE statements. This is formed in the same way as that of the control group, except the 

interaction term differs. For the interaction term, insert the continuous time value into the first 

position to achieve ‘group*t 1 0’. When checking, we see that the ESTIMATE output in Figure 3 

matches the calculations from the formulation above using the parameter estimates in Figure 2. 

3.1.4 CHANGE DIFFERENTLY BY GROUP 

To find the difference in slopes between the FACT intervention and control groups, consider the 

above slopes for the control and intervention groups separately and subtract them. The difference in 

slopes between the FACT intervention and control groups can be derived:  

= (𝛽2 +  𝛽3) −  𝛽2 

=  𝛽3 

= 4.4 

        The necessary ESTIMATE statement is: 

ESTIMATE "diff in slope" group * t 1 -1; 

To write the statement for the difference in slopes between the FACT intervention and control 
groups, the interaction term is involved. When comparing two groups in SAS ESTIMATE statements 
1 and -1 are used. When checking, we see that the ESTIMATE output in Figure 3 matches the 
calculations from the formulation above using the parameter estimates in Figure 2. 

3.2 SUMMARY OF RESULTS OUTPUT 

 
Figure 3. ESTIMATE statement results. 

 
 

Using the model-based estimates at each timepoint, along with the confidence intervals, a summary 

graphic can be made to show summary changes over time (Figure 4).   

  



Figure 4. ESTIMATE statement results for mean CSES over time in cancer patients receiving 
FACT intervention vs. the control. 

 

 

4 CATEGORICAL TIME 

When treating time as a categorical variable, the model can be written as:  

𝑔(𝜇𝑖𝑡) = 𝑎 + 𝛽1𝑔𝑟𝑜𝑢𝑝 + 𝛽2𝑡𝑖𝑚𝑒1 + 𝛽3𝑡𝑖𝑚𝑒2 +  𝛽4𝑔𝑟𝑜𝑢𝑝0 ∗ 𝑡𝑖𝑚𝑒1 + 𝛽5𝑔𝑟𝑜𝑢𝑝0 ∗ 𝑡𝑖𝑚𝑒2 +  𝛽6𝑔𝑟𝑜𝑢𝑝1 ∗ 𝑡𝑖𝑚𝑒0

+ 𝛽7𝑔𝑟𝑜𝑢𝑝1 ∗ 𝑡𝑖𝑚𝑒1 + 𝛽8𝑔𝑟𝑜𝑢𝑝1 ∗ 𝑡𝑖𝑚𝑒2 
 

Where g(.) is the specified link function, 𝜇𝑖𝑡 is the mean CSES at time t for subject i. The following SAS 
code represents the model, where reference level coding is utilized:  

proc genmod data=dataset; 

  class group(ref="Control (waitlist)") t(ref="0") index;    

   model outcome = group t group*t/ dist=normal link=identity; 

     repeated subject=index / type=exch; 

run; 

 

When running the code above, the following output is produced detailing the parameter information and 
the parameter estimates (Figure 5 and Figure 6). The only item that differs in the PROC GENMOD 
statement from treating time as continuous is adding the time variable to the class statement. Using the 
equation derived in the next section, the parameter estimates can be used to check the ESTIMATE 
statement results.  

  



Figure 5. Parameter information for model with categorical time. 

 

Figure 6. Parameter estimates after running the PROC GENMOD procedure with categorical time. 

 

4.1 COMPARISONS OF INTEREST 

We can address the same four main research questions for this scenario via ESTIMATE statements: 
1) What are the model estimated outcome measure scores at each time point for each group?; 2) Are 
the outcome measure scores different by group at any timepoint?; 3) Is there a significant change over 
time in outcome measure within group (separately for control and FACT intervention)?; and 4) Does 
the outcome measure change differently over time by group? (i.e. is there a difference in slopes 
between the FACT intervention and control groups?). 

  



4.1.1 ESTIMATED SCORES AT EACH TIMEPOINT 

CONTROL: Model estimates for the mean CSES score for the control group at each time point 

Estimate for control at time=0; Let group = 0, time = 0 
 

𝑔(𝜇𝑖𝑡) = 𝑎 + 𝛽1(0) + 𝛽2(0) + 𝛽3(0) +  𝛽4(0)(0) + 𝛽5(0)(0) + 𝛽6(0)(0) + 𝛽7(0)(0) + 𝛽8(0)(0) 

= 𝑎 

= 79.1 

 

The ESTIMATE statement is: 

ESTIMATE "model est control at t0" int 1  group 0 1  t 0 0 1  

  group*t 0 0 0 0 0 1; 

To write the statement for the control group at time 0, we include the intercept, time, group, and the 

group*time interaction term in the ESTIMATE statements. The intercept is included as ‘int 1’, derived 

directly from the linear combinations in the equations above. In Figure 5, the parameter information 

is ordered, where the FACT intervention group is listed first as Prm2, in comparison to Prm3 

representing the control group (reference). Therefore, to indicate the control group, we use the 

second position of the group section, giving ‘group 0 1.’ Similarly, the parameter order for the time 

variable is important for formulating the ESTIMATE statement such that Prm4 represents time 1, 

Prm5 represents time 2, and Prm6 represents t0, thus implying ‘t 0 0 1’. The interaction term, 

group*t, has 6 positions to fill in the ESTIMATE statement. This is achieved by the 2 study groups 

(FACT intervention vs. control) multiplied by 3 time points (t0, t1, t2). Based on Figure 5, Prm12, the 

last of the 6 interaction parameters, represents the interaction for the control group*t0, thus ‘group*t 

0 0 0 0 0 1’. When checking, we see that the ESTIMATE output in Figure 8 matches the calculations 

from the formulation above using the parameter estimates in Figure 6.  

 

The following ESTIMATE statements extend the example to obtain model estimates for the control 

group at times 1 and 2:  
ESTIMATE "model est control at t1" int 1  group 0 1  t 1 0 0  

group*t 0 0 0 1 0 0;  

ESTIMATE "model est control at t2" int 1  group 0 1  t 0 1 0  

group*t 0 0 0 0 1 0;   

 

INTERVENTION: Model estimates for the mean CSES score for the FACT intervention group at 
each time point: 

Estimate for FACT intervention group at time=1; Let group = 1, time = 1 
𝑔(𝜇𝑖𝑡) = 𝑎 + 𝛽1(1) +  𝛽2(1) + 𝛽3(0) +  𝛽4(0) ∗ (1) + 𝛽5(0)(0) +  𝛽6(1)(0) + 𝛽7(1)(1) + 𝛽8(1)(0) 

= 𝑎 + 𝛽1 + 𝛽2 + 𝛽7 
= 79.1 + 1.3 + 1.9 + 4.0 

= 86.3 
The ESTIMATE statement is: 

ESTIMATE "model est FACT at t1" int 1  group 1 0  t 1 0 0  

group*t 1 0 0 0 0 0; 

To write the statement for the FACT intervention at time 1, we include the intercept, time, group, and 

the group*time interaction term in the ESTIMATE statements. The intercept is included as ‘int 1’, 

derived directly from the linear combinations in the equations above. In Figure 5, the parameter 

information is ordered, where the FACT intervention group is listed first as Prm2, in comparison to 

Prm3 representing the control group (reference). Therefore, to indicate the FACT intervention group, 



we use the first position of the group section, giving ‘group 1 0.’ Similarly, the parameter order for the 

time variable is important for formulating the ESTIMATE statement such that Prm4 represents time 

1, Prm5 represents time 2, and Prm6 represents t0, thus implying ‘t 1 0 0’. The interaction term, 

group*t, has 6 positions to fill in the ESTIMATE statement. Based on Figure 5, Prm7, is the relevant 

interaction parameter representing the interaction for the FACT intervention group*t1, thus ‘group*t 1 

0 0 0 0 0’. When checking, we see that the ESTIMATE output in Figure 8 matches the calculations 

from the formulation above using the parameter estimates in Figure 6.  

 

A helpful tool is the /e tool on the ESTIMATE statement to check coefficients for parameters. The 

following code produces the output below (Figure 7): 

 
ESTIMATE "model est FACT at t1" int 1  group 1 0  t 1 0 0  

group*t 1 0 0 0 0 0/e; 

 

Figure 7. Coefficients for ESTIMATE statements. 

 
 

The following ESTIMATE statements extend the example to obtain model estimates for the FACT 

intervention group at each time 0 and 2:  
ESTIMATE "model est FACT at t0" int 1  group 1 0  t 0 0 1  

group*t 0 0 1 0 0 0; 

ESTIMATE "model est FACT at t2" int 1  group 1 0  t 0 1 0  

group*t 0 1 0 0 0 0; 

4.1.2 DIFFERENCES IN SCORES AT EACH TIMEPOINT 

Are the outcome measure scores different by group at any time (for example, at time 2)?  

• Let group = 1, time = 2 
𝑔(𝜇𝑖𝑡) = 𝑎 + 𝛽1(1) +  𝛽2(0) + 𝛽3(1) +  𝛽4(0)(0) + 𝛽5(0)(1) + 𝛽6(1)(0) + 𝛽7(1)(0) + 𝛽8(1)(1) 

= 𝑎 + 𝛽1 + 𝛽3 + 𝛽8 

= 79.1 + 1.3 + 0.5 + 9.0 

• Let group = 0, time = 2 
𝑔(𝜇𝑖𝑡) = 𝑎 + 𝛽1(0) +  𝛽2(0) + 𝛽3(1) + 𝛽4(1)(0) +  𝛽5(1)(1) +  𝛽6(0)(0) + 𝛽7(0)(0) + 𝛽8(0)(1) 

= 𝑎 + 𝛽3 + 𝛽5 

= 79.1 + 0.5 + 0 

To find the difference between these two groups we subtract:  
= (𝑎 + 𝛽1 + 𝛽3 + 𝛽8) − (𝑎 + 𝛽3 + 𝛽5) 

= 𝛽1 + 𝛽8 − 𝛽5 

= 1.3 + 9.0 + 0 

= 10.3 

𝛽1 + 𝛽8 − 𝛽5 represents the difference between FACT at time=2 and control at time=2 

 

The necessary ESTIMATE statement is: 
ESTIMATE "Diff FACT v Control at t2" group 1 -1 group*t 0 1 0 0 -1 0; 

 



To write the statement for the difference in average CSES between the FACT intervention and 
control groups at time 2, we include the group, and the group*time interaction term in the ESTIMATE 
statements. Note that 𝛽1 + 𝛽8 − 𝛽5 correspond to these terms in the model. In Figure 5, the 
parameter information is ordered, where the FACT intervention group is listed first as Prm2, in 
comparison to Prm3 representing the control group (reference). In SAS, when a 1:1 comparison is 
made between two groups, the syntax is 1 for the main group and -1 for the comparator group, thus 
‘group 1 -1’ is utilized to represent the comparison between the FACT intervention group and control 
group. Similarly, the interaction term follows the same syntax indicating a comparison, Based on 
Figure 5, Prm8 and Prm11 are the relevant interaction parameters, thus ‘group*t 0 1 0 0 -1 0’. When 
checking, we see that the ESTIMATE output in Figure 8 matches the calculations from the 
formulation above using the parameter estimates in Figure 6. 

While these differences can also be achieved with the LSMEANS statement, lsmeans group*t / 

cl diff;, the comparisons are produced for all combinations of group and time, thus cluttering the 

output with comparisons that are irrelevant to compare in clinical practice. It is useful to be able to 
logically write the estimate statements to simplify the output produced by the GEMOD procedure.  

The following ESTIMATE statements extend the example to obtain model estimates for the 

difference in mean CSES estimate for the FACT intervention group vs. the control group at time 0 

and time 1. 
ESTIMATE "Diff FACT vs. Control at t0" group 1 -1 group*t 0 0 1 0 0 -1; 

ESTIMATE "Diff FACT vs. Control at t1" group 1 -1 group*t 1 0 0 -1 0 0; 

 

4.1.3 CHANGE OVER TIME (BETWEEN CERTAIN TIMEPOINTS) WITHIN GROUP 

CONTROL: For the control group, group=0. We want to compare the change from t0 to t1. From the 
model above, we can write the following:   

• Let group = 0 and time = 1  
= 𝑎 + 𝛽2 

=  79.1 + 1.9 

• Let group = 0 and time = 0 
= 𝑎 

= 79.1 

To find the difference between these two groups we subtract:  
= (𝑎 + 𝛽2) − (𝑎) = 𝛽2 

= 1.9 

𝛽2 represents the difference between control at time=1 and control at time=0. (i.e. slope for control) 
The necessary ESTIMATE statement is: 
ESTIMATE 'slope for Control t0-t1' t 1 0 -1 group*t 0 0 0 1 0 -1; 

  

To write the statement for the change over time in average CSES in the control group between time 

0 and time 1, we include the time and the group*time interaction term in the ESTIMATE statements. 

In Figure 5, the parameter information is ordered, where the FACT intervention group is listed first as 

Prm2, in comparison to Prm3 representing the control group (reference). Similarly, the parameter 

order for the time variable is Prm4(time 1), Prm5(time 2), and Prm6(time 0). Since a comparison is 

being made between time 0 and time 1, the appropriate syntax is ‘t 1 0 -1’. For the interaction terms, 

insert the relevant parameters are Prm10 and Prm12, to achieve ‘group*t 0 0 0 1 0 -1’. When 

checking, we see that the ESTIMATE output in Figure 8 matches the calculations from the 

formulation above using the parameter estimates in Figure 6. 

 

Similarly, for the difference between control at time=2 and time=1 
ESTIMATE 'slope for Control t1-t2' t -1 1 0 group*t 0 0 0 -1 1 0; 

 



INTERVENTION: For the FACT intervention group, group=1, and we want to compare the change 

from t0 to t1.  

 

• Let group = 1, time = 1  
= 𝑎 + 𝛽1 +  𝛽2 +  𝛽7 

= 79.1 + 1.3 + 1.9 + 4.0 

• Let group = 1, time = 0 
= 𝑎 + 𝛽1 +  𝛽6 

= 79.1 + 1.3 + 0 

To find the difference between these two groups we subtract:  
= (𝑎 + 𝛽1 + 𝛽2 + 𝛽7) − (𝑎 + 𝛽1 + 𝛽6) 

= 𝛽2 +  𝛽7 − 𝛽6 

= 1.9 + 4.0 − 0 

= 5.9 

𝛽2 +  𝛽7 − 𝛽6 represents the difference between FACT at time=1 and FACT at time=0. (i.e. slope for 
FACT)  
 

The necessary ESTIMATE statement is: 
ESTIMATE 'slope for FACT t0-t1' t 1 0 -1 group*t 1 0 -1 0 0 0; 

 

To write the statement for the change over time in average CSES in the FACT intervention group 

between time 0 and time 1, we include the time and the group*time interaction term in the 

ESTIMATE statements. This is formed in the same way as that of the control group, except the 

interaction term differs. For the interaction term, insert the appropriate interaction parameters (Prm7 

and Prm9) to achieve ‘group*t 1 0 -1 0 0 0’. When checking, we see that the ESTIMATE output in 

Figure 8 matches the calculations from the formulation above using the parameter estimates in 

Figure 6. 

 

Similarly, for the difference between the FACT intervention at time 2 and time 1: 
ESTIMATE 'slope for FACT t1-t2' t -1 1 0 group*t -1 1 0 0 0 0; 

4.1.4 CHANGE DIFFERENTLY BY GROUP 

To find the difference in slopes between the FACT intervention and control groups, consider the 

above slopes for the control and intervention groups separately at time 0 and time 1 and subtract 

them. The difference in slopes between the FACT intervention and control groups can be derived.  

(𝛽2 + 𝛽7 − 𝛽6) −  𝛽2 

=  𝛽7 − 𝛽6 

= 4.0 

        The necessary ESTIMATE statement is: 

ESTIMATE "diff in slope t0-t1" group*t 1 0 -1 -1 0 1; 

To write the statement for the difference in slopes between the FACT intervention and control 
groups, the interaction term is involved. When comparing two groups in SAS ESTIMATE statements 
1 and -1 are used. Since we are making two inherent comparisons and finding the difference of 
slopes, the interaction parameters for both the comparison between outcome scores for control and 
the comparison between outcome scores for the FACT intervention from time 0 to time 1 need to be 
included as comparisons in the ESTIMATE statement. This results in utilizing Prm7, Prm 9, Prm10, 
and Prm12 from Figure 5 to give group*t ‘1 0 -1 -1 0 1’. When checking, we see that the ESTIMATE 
output in Figure 8 matches the calculations from the formulation above using the parameter 
estimates in Figure 6. 



Similarly, the difference in slopes between the FACT intervention and control groups at time 1 and 

time 2 is as follows:  
ESTIMATE "diff in slope t1-t2" group*t -1 1 0 1 -1 0; 

 

4.2 SUMMARY OF RESULTS OUTPUT 

For all examples above involving categorical time, the ESTIMATE output in Figure 8 matches the 
calculations from the formulas above using the parameter estimates in Figure 6. Note that all model 
estimates are similar to estimates in section 3 when treating time as continuous.  

Figure 8. ESTIMATE statement results for categorical time 

 

  



 

Using the model-based estimates at each timepoint, along with the confidence intervals, a summary 

graphic can be made to show summary changes over time (Figure 9).   

 

Figure 9. ESTIMATE statement results for mean CSES over time in cancer patients receiving 

FACT intervention vs. the control (categorical time). 

 

5 CONCLUSION 

Often in cancer research, continuous outcomes, such as survey instrument scores, are measured 
repeatedly over time. A common method for appropriately handling correlated outcome data collected 
over time is the usage of marginal (linear) models, with generalized estimating equation (GEE) methods. 
This is a powerful tool; however, the ESTIMATE statements utilized to discern model estimated outcome 
scores are not easily understood. By using the underlying model structure and equations to stencil the 
model parameters and double check estimates produced, the ESTIMATE statements are demystified and 
more clearly illustrated. This allows for more useful and straightforward output.  

8 APPENDICES 

8.1 GENMOD FOR CONTINUOUS WITH ALL ESTIMATE STATEMENTS 

 
proc genmod data=dataset; 

  class group(ref="Control (waitlist)") index;     

   model outcome = group t group*t/ dist=normal link=identity; 

     repeated  subject=index / type=exch; 

   

  /*1*/ 

  ESTIMATE "model est control at t0"  int 1  group 0 1  t 0  

group*t 0 0; 

   ESTIMATE "model est control at t1"  int 1  group 0 1  t 1  

group*t 0 1; 

    ESTIMATE "model est control at t2"  int 1  group 0 1  t 2  

group*t 0 2;  

  ESTIMATE "model est FACT at t0"  int 1  group 1 0  t 0  

group*t 0 0; 

ESTIMATE "model est FACT at t1"  int 1  group 1 0  t 1  

group*t 1 0;  

ESTIMATE "model est FACT at t2"  int 1  group 1 0  t 2  

group*t 2 0;  

 

  /*2*/ 

  ESTIMATE "Diff FACT vs. control at t0" group 1 -1 ; 

  ESTIMATE "Diff FACT vs. control at t1" group 1 -1 group*t 1 -1; 

  ESTIMATE "Diff FACT vs. control at t2" group 1 -1 group*t 2 -2; 

 

  /*3*/ 

  ESTIMATE "slope for Control"  t 1 group*t  0 1 ; 

  ESTIMATE "slope for FACT"  t 1 group*t  1 0 ; 

 

  /*4*/ 

  ESTIMATE "diff in slope" group*t  1 -1 ; 

 run; 

  



8.2 GENMOD FOR CATEGORICAL 

proc genmod data=dataset; 

  class group(ref="Control (waitlist)") t(ref="0") index;   

   model outcome = group t group*t/ dist=normal link=identity; 

     repeated subject=index / type=exch; 

 

  /*1*/ 

  ESTIMATE "model est Control at t0"  int 1  group 0 1  t 0 0 1  

group*t 0 0 0 0 0 1; 

  ESTIMATE "model est Control at t1"  int 1  group 0 1  t 1 0 0  

group*t 0 0 0 1 0 0;  

ESTIMATE "model est Control at t2"  int 1  group 0 1  t 0 1 0  

group*t 0 0 0 0 1 0;   

  ESTIMATE "model est FACT at t0"  int 1  group 1 0  t 0 0 1  

group*t 0 0 1 0 0 0; 

  ESTIMATE "model est FACT at t1"  int 1  group 1 0  t 1 0 0  

group*t 1 0 0 0 0 0;  

  ESTIMATE "model est FACT at t2"  int 1  group 1 0  t 0 1 0  

group*t 0 1 0 0 0 0; 

 

  /*2*/ 

  ESTIMATE "Diff FACT vs. Control at t0" group 1 -1  

group*t 0 0 1 0 0 -1 ; 

  ESTIMATE "Diff FACT vs. Control at t1" group 1 -1  

group*t 1 0 0 -1 0 0; 

  ESTIMATE "Diff FACT vs. Control at t2" group 1 -1  

group*t 0 1 0 0 -1 0; 

 

  /*3*/ 

  ESTIMATE 'slope for Control t0-t1' t 1 0 -1 group*t 0 0 0 1 0 -1; 

  ESTIMATE 'slope for FACT t0-t1'    t 1 0 -1 group*t 1 0 -1 0 0 0; 

  ESTIMATE 'slope for Control t1-t2' t -1 1 0 group*t 0 0 0 -1 1 0; 

  ESTIMATE 'slope for FACT t1-t2'    t -1 1 0 group*t -1 1 0 0 0 0; 

 

  /*4*/ 

  ESTIMATE "diff in slope t0-t1" group*t  1 0 -1 -1 0 1; 

  ESTIMATE "diff in slope t1-t2" group*t  -1 1 0 1 -1 0; 

 run; 

 


