
Introduction



What are Patient Profiles?
• Individual reports for each subject in a clinical study

• Contain all data collected, or a specific subset

• Typically raw data – no summary, derivations or analysis

• Used for data cleaning and study monitoring



Design Objectives
• Input: SDTM data sets

• Output: One PDF file per subject

• Can be rerun iteratively as data is collected

• No hard-coding subject numbers – must be dynamic!

• Include both tabular and graphical output

• Provide output data sets for independent QC

• Use color coding to highlight changes since prior run



Basic subject information in titles

Key variables from Demographics domain

List of subject visits and dates

Detail report of adverse events

Orange row is new data 
since the prior run. 

Yellow cells have changed 
since the prior run.



Vital signs in tabular format:
One row per parameter,

Visits transposed to columns

Vital signs in graphical format:
One panel per parameter,

Visits across X axis



This programming framework allows you to: 
• Customize titles and footnotes per subject

• Include desired data elements in any order

• Paginate as desired

• Mix both tabular and graphical output

• Customize colors, fonts, borders, etc. using ODS

• Present data both:
◦ Vertically – list individual records (adverse events, medical history, etc.)

◦ Horizontally – transposed by visit (labs, vital signs, ECGs, etc.)

Everything is done using the 
base SAS 9.4 product.



Loading the Data



The Section List
• The list of SDTM domains to be included is stored in a macro 

variable:

• These domains will appear in the output in the order listed.

• For each domain, we must create two macros:
◦ m_getdata_XX (to load the data and perform any manipulations)

◦ m_printdata_XX (to generate tabular and/or graphical output)

• Demographics (DM) must be included - it is where we get the list 
of subjects (more on that later)

%let sectionlist = DM SV AE VS MH CM QS LB EG;



The Get Data Loop

%do secnum = 1 %to %sysfunc(countw(&sectionlist));

   %let sec = %scan(&sectionlist, &secnum);

   %put === Getting Data for &sec section ===;

   %m_getdata_&sec;

%end;

Macro %DO loop executes once 
for each section in the list.

Get the corresponding section 
name from the list.

A custom log message makes 
the execution easier to trace.

Call the macro to load the data.



A Simple Get Data Macro

%macro m_getdata_dm;

   data dmfinal;

      set sdtm.dm;

      %if &subj_subset ne %then where &subj_subset;;

   run;

%mend m_getdata_dm;

Optional subsetting criteria are 
placed into another macro 

variable defined at the top of 
the main program.

Two semicolons are needed here. 
One ends the %IF macro statement 

and the other ends the WHERE 
statement being generated.



A More Complex Get Data Macro
%macro m_getdata_ae;

  /* Omitted: code to merge SDTM.AE and SDTM.SUPPAE */

  data aefinal;

    set aemerge;

    %if &subj_subset ne  %then where &subj_subset;;

    length _aeterm $200 aestart aeend $20 _aeser $3 _aesev $8 _aerel $8

          _aeout $200  aeacn $30 _aetrtem $3;

    _aeterm   = catx(': ', AESPID, AETERM);

    aestart   = catx(' ', AESTDTC, ifc(AESTDY, cats('(', AESTDY, ')'), ''));

    aeend     = catx(' ', AEENDTC, ifc(AEENDY, cats('(', AEENDY, ')'), ''));

    _aeser    = put(AESER, $yn.);

    _aesev    = propcase(AESEV);

    _aerel    = propcase(AEREL);

    _aeout    = propcase(AEOUT);

    _aeacn    = propcase(AEACN);

    _aetrtem  = put(AETRTEM, $yn.);

    if not nmiss(AESTDY, AEENDY) then aedur = AEENDY - AESTDY + 1;

  run;

%mend m_getdata_ae;

Custom variables are created 
to display data as desired.



Setting Up the
 Subject List



Creating the Profile List Data Set

proc sql;

  create table profilelist as

  select a.USUBJID label = 'Unique Subject ID',

    a.SUBJID label = 'Subject ID',

    /* Additional subject-level fields to appear in header */

  from (select * from dmfinal where ACTARMCD not in ('Scrnfail','')) a

    left join

      /* Join with additional SDTM data sets as needed */

    on a.USUBJID = b.USUBJID;

quit;

PROFILELIST data set contains one row per subject and includes subject-
level fields that will appear in the profile header. These are typically things 
like treatment status, date of first dose, study status, date of last visit, etc.



Generating the Table of Contents

ods excel file="&outpath.\patient_profile_master_list_&rundate..xlsx"

          options (sheet_name = "Patient Profile List");

proc report data = profilelist;

 columns usubjid subjid /* Additional subject-level fields */;

run;

ods excel close; Optionally, we can write the contents of the 
PROFILELIST data set to an external file to serve as 

a Table of Contents for our patient profiles.



Profile list 

output as 

an Excel 

workbook



Building the Macro Variable Lists

proc sql noprint;

   select distinct USUBJID, SUBJID

     into :usubj1 - , :subj1 -

   from profilelist

   order by USUBJID;

   %let subjcount = &sqlobs;

 quit;

Create two series of macro variables:
 USUBJ1, USUBJ2, … contains USUBJID for each subject

SUBJ1, SUBJ2, … contains SUBJID for each subject

%put _user_;

GLOBAL SUBJ1 1015
GLOBAL SUBJ2 1023
GLOBAL SUBJ3 1028
GLOBAL SUBJ4 1033
...
GLOBAL SUBJCOUNT 254
...
GLOBAL USUBJ1 01-701-1015
GLOBAL USUBJ2 01-701-1023
GLOBAL USUBJ3 01-701-1028
GLOBAL USUBJ4 01-701-1033



Building the Outputs



The Print Data Loop (Slide 1 of 2)

%do i = 1 %to &subjcount;

   proc sql noprint;

     select coalescec(firstdose, 'No Data'),

       subjstatus, lastvis, visabbr

       into

       :firstdose, :subjstatus, :lastvis, :visabbr trimmed

     from profilelist

     where USUBJID = "&&usubj&i";

   quit;

The %DO macro loop executes 
once for each subject.

The header values for the current 
subject are placed into macro variables.



The Print Data Loop (Slide 2 of 2)

ods pdf file = "&outpath\profile_&&subj&i.._&visabbr._&rundate..pdf"

              startpage=no nogtitle nogfootnote;

  title1 j = l "Study: CDISCPILOT01"

         j = r "Subject: &&usubj&i (&&subj&i)";

  /* Additional titles and footnotes as desired */

  %do secnum = 1 %to %sysfunc(countw(&sectionlist));

    %let sec = %scan(&sectionlist, &secnum);

    %put === Printing Data for &sec for Subject &&usubj&i (&&subj&i) ===;

    %m_printdata_&sec;

  %end;

  ods pdf close;

%end;

We’re still inside the %DO loop!

Open a PDF for the 
current subject.

Inner %DO loop prints each 
section of the report.



Vertical Tabular Output
%macro m_printdata_ae;

  proc sql noprint;

    select count(*) into :numaerecs from aefinal where USUBJID="&&usubj&i";

  quit;

  %if &numaerecs > 0 %then %do;

    proc report data=aefinal split='~' style(report)=[just=left];

      where USUBJID="&&usubj&i";

      columns AESPID ('Adverse Events' _aeterm aedtc aestart aeend

        _aeser aedur _aesev _aerel _aeout _aeacn _aetrtem);

      /* DEFINE statements */

    run;

  %end;

%mend m_printdata_ae;

Only print this section of the report if 
the current subject has data!

PROC REPORT subsets for the 
current subject and displays 

the columns we choose.



Horizontal Tabular Output (Slide 1 of 2)
%macro m_printdata_vs;

  /* Omitted: code for selecting and formatting data */

 %if &numvsrecs > 0 %then %do;

  /* Omitted: sort data for transposing */

  proc transpose data = vssubset out = vstrans(drop = _:);

    by VSTESTCD;

    var VSSTRESN;

    id vislbl;

  run;

Transpose visits to columns. VISLBL is 
a derived variable containing the 

desired column labels for each visit. 



Horizontal Tabular Output (Slide 2 of 2)
proc sql noprint; 

    select distinct NAME, input(NAME, visord.) as sort1

      into :vitalsvislist separated by ' ', :dummy1

    from DICTIONARY.COLUMNS

    where LIBNAME = 'WORK' and MEMNAME = 'VSTRANS' and 

      /* Criteria for selecting visit columns */

      order by sort1, sort2;

    %let numvitalsvisits = &sqlobs;

  quit;

  proc report data = vstrans split = '~' style(report) = [just = left];

    columns dummy ('Vital Signs' VSTESTCD &vitalsvislist);

    /* DEFINE statements */

  run;

%end;

%mend m_printdata_vs;

We don’t want to 
hardcode the visits, 
so we place the list 

of visits into a macro 
variable.

The list of visits could vary by 
subject or even by report section!



Graphical Output
proc sql noprint;

    select min(xvar), max(xvar) into :xmin, :xmax

    from svfinal

    where VISITNUM = int(VISITNUM) and USUBJID = "&&usubj&i";

  quit;

  /* Omitted: code for removal of unscheduled visits and block rendering */

  proc sgpanel noautolegend data = vsfigure dattrmap = vsattrmap;

    format xvar xvarfmt. VSTESTCD $vsfmt.;

    panelby VSTESTCD / onepanel uniscale = column novarname 

                       layout = rowlattice headerattrs = (size = 8pt);

block x = xvar block = blocklbl / /* Additional formatting options */;

    series x = xvar y = VSSTRESN    / /* Additional formatting options */;

    rowaxis display = (nolabel) thresholdmax=1 thresholdmin=1 offsetmax=0.1;

    colaxis label = 'Visit' values = (&xmin to &xmax by 1) fitpolicy=staggerthin;

  run;

Dynamically determine 
X axis range

SGPANEL creates a separate plot for each parameter.



Additional Features



Creating Validation Data Sets (1 of 2)
• During the Get Data loop, we initialize an empty QC data set for 

each section of the report:

data PDATA.PP_&sec;

   length USUBJID $23;

   call missing(USUBJID);

   stop;

 run;



Creating Validation Data Sets (2 of 2)
• We create an output data set from PROC REPORT each time it is 

called and append to the QC data set for that report section.

proc report data=aefinal split='~' style(report)=[just=left] 

out=work.qc_ae_&i;

   ...

run;

data PDATA.PP_AE;

  set PDATA.PP_AE

      work.qc_ae_&i(in=currsubj where=(_break_=''));

  if currsubj then USUBJID="&&usubj&i";

  drop dummy _break_;

run;



Highlighting New/Changed Data
• To facilitate data review, changes and additions since the 

previous run can be highlighted in different colors

• During the Get Data loop, an additional macro is called that adds 
two variables to each data set:
◦ newflag – a 0 or 1 flag to indicate whether a record is new

◦ modcols – a list of variables that have different values from the prior run

• During the Print Data loop, additional logic is added to each 
PROC REPORT to modify the attributes based on these variables



Highlighting New/Changed Data
%macro m_add_update_vars(dsetin_curr=, dsetin_prev=, dsetout=, 

  keyvarlist=, othvarlist= );

  /* Omitted: code to sort input data sets */

  data &dsetout;

    merge &_prefix._curr ( in = incurr)

          &_prefix._prev ( in = inprev keep = &keyvarlist &othvarlist rename = ( 

    %do _i=1 %to %sysfunc(countw(&othvarlist,%str( )));

     %scan(&othvarlist,&_i) = _%scan(&othvarlist,&_i) %end;));

    by &keyvarlist; if incurr;

 length modcols $200; call missing(modcols);

 if not(inprev) then newflag=1;

 else do;

  %do _i=1 %to %sysfunc(countw(&othvarlist,%str( )));

   if %scan(&othvarlist,&_i) ne _%scan(&othvarlist,&_i) then modcols =

            catx(' ',modcols,"%upcase(%scan(&othvarlist,&_i))"); %end;

 end; 

run;

%mend m_add_update_vars;
Merge together current and prior data sets, check for new 

records, and compare variables across existing records.



Highlighting New/Changed Data
proc report data=aefinal split='~' style(report)=[just=left] out=work.qc_ae_&i;

  where USUBJID="&&usubj&i";

  columns AESPID /* additional variables */

    %if %nrbquote(&highlight_updates)=Y %then newflag modcols;;

  /* DEFINE statements */

  %if %nrbquote(&highlight_updates)=Y %then %do;

    define newflag   / display noprint;

    define modcols   / display noprint;

  %end;

  compute _aetrtem;

    %if %nrbquote(&highlight_updates)=Y %then %do;

    if newflag=1 then call define(_row_, "style","style=[background=lightorange]");

      do i=1 to countw(modcols);

        call define(scan(modcols, i), "style", "style=[background=lightyellow]");

      end;

    %end;

  endcomp;

run;

A global macro variable 
HIGHLIGHT_UPDATES is used to 
turn the highlighting on or off.

When highlighting is enabled, the COLUMNS statement is modified 
and additional DEFINE statements and a COMPUTE block are added.



Adding PDF Bookmarks
PDF bookmarks 
make it easier to 
navigate the patient 
profile output, 
particularly when 
the report contains 
many sections.



Adding PDF Bookmarks
• Getting the bookmarks just right requires a few adjustments:

◦ On the ODS PDF statement, change the behavior with the PDFTOC=1 option.

◦ Use ODS PROCLABEL to override the bookmark name for each PROC REPORT.

◦ Use the CONTENTS=‘ ‘ option on PROC REPORT to suppress second level 
bookmarks

◦ Use a BREAK statement with a dummy variable on each PROC REPORT to 
suppress third level bookmarks.





Conclusion



Summary
• Patient profiles are a useful tool for clinical data review and 

monitoring.

• The programming framework presented provides a flexible and 
customizable way to create highly customized patient profiles.

• The techniques used adapt to whatever subjects and visits appear 
in the data without the need for code modifications.



Possible Future Enhancements
• Use of a control file to specify:

◦ Specify which subjects to include

◦ Specify which data sections in include and their order

◦ Allow for different sections to be specified for different subjects

◦ Control grouping of parameters into pages

