
Abstract 

Joining two or more tables of data is a powerful feature in the relational model. The SQL procedure enables users 

to join tables of information quickly and easily. Linking one piece of information with another piece of information 

is made possible when at least one column is common to each table. This paper describes several join types 

including the importance of joins, the difference between the various join types (inner and outer), the creation of 

outer (left, right, and full) joins, the importance of the WHERE / ON clause(s) in creating and subsetting rows of 

data, the purpose of creating and using table aliases to minimize the number of keystrokes, and the application of 

set operators. 

 

Introduction 

A powerful and essential PROC SQL programming technique that all SAS users should be aware of, and comfortable 

performing, is the process of joining (or combining) two or more tables of data. Specific topics include describing the 

join process, what a join is, the preparation requirements for each table being specified in a join, the join syntax, the 

various types of joins available to SAS users, conventional (symmetrical) and unconventional PROC SQL join 

programming techniques essential concepts, syntax, and programming techniques. 

 

Example Tables 

The examples used throughout this paper utilize a database of two tables. (A relational database is a collection of 

tables.) The data used in all the examples in this paper consists of a selection of movies that I’ve viewed over the 

years. The Movies table consists of six columns: title, length, category, year, studio, and rating. Title, category, 

studio, and rating are defined as character columns with length and year being defined as numeric columns. The 

data stored in the Movies table is depicted below. 

MOVIES Table 

The data stored in the ACTORS table consists of three columns: title, actor_leading, and actor_supporting, all of 

which are defined as character columns. The data stored in the Actors table is illustrated below. 

ACTORS Table 

 
 

Cross Join or Cartesian Product 

A CROSS JOIN or Cartesian Product represents all rows in Table A multiplied by all rows in Table B. Due to this 

multiplier effect, the results can be quite large. The results are shown in the shaded area in the Venn diagram. 

 

 

 

LIBNAME MYDATA "/home/kirklafler/Data Sources" ; 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M, 

       MYDATA.Actors A ; 

QUIT ; 

 

Results for Inner Join: 

 
 

. . . . .   . . . . .  . . . . . .  . . . . .  . . . . .   . . . . .  



Visualizing SQL Join Types with Venn Diagrams, continued 

 
 

 

Inner Join 

An INNER JOIN joins (or merges) ONLY the matching rows in BOTH tables. Results are found in the overlapping area 

as shown in the Venn diagram. 

 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M 

         INNER JOIN 

       MYDATA.Actors A 

   ON M.Title = A.Title ; 

QUIT ; 

 

Results for Inner Join: 

Left Outer Join 

A LEFT OUTER JOIN returns all rows from the left table (Table A) with the matching rows from the right table (Table 

B) or null as shown in the Venn diagram. 

 

 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M 

         LEFT JOIN 

       MYDATA.Actors A 

   ON M.Title = A.Title ; 

QUIT ; 

 

 

Results for Left Outer Join: 

 
 

 

Right Outer Join 

A RIGHT OUTER JOIN returns all rows from the right table (Table B) with the matching rows from the left table 

(Table A) or null as shown in the Venn diagram. 

 

 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M 

         RIGHT JOIN 

       MYDATA.Actors A 

   ON M.Title = A.Title ; 

QUIT ; 

Results for Right Outer Join: 

 
 

 

Full Outer Join 

A FULL OUTER JOIN returns matched and unmatched rows from both tables (it's essentially a union of both tables) 

as shown in the Venn diagram. 

 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M 

         FULL JOIN 

       MYDATA.Actors A 

   ON M.Title = A.Title ; 

QUIT ; 

 

Results for Full Outer Join: 

Outer Excluding Join 

An OUTER EXCLUDING JOIN returns all rows in Table A and all rows in Table B that don't match as shown in the 

Venn diagram. 

 

 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M 

         FULL JOIN 

       MYDATA.Actors A 

   ON M.Title = A.Title 

    WHERE M.Title IS NULL 

       OR A.Title IS NULL ; 

QUIT ; 

 

 

Results for Outer Excluding Join: 

 
 

 

Left Excluding Join 

A LEFT EXCLUDING JOIN returns all rows in Table A that don't match any rows in Table B as shown in the Venn 

diagram. 

 

 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M 

         LEFT JOIN 

       MYDATA.Actors A 

   ON M.Title = A.Title 

    WHERE A.Title IS NULL ; 

QUIT ; 

Results for Left Excluding Join: 

 
 

 

Right Excluding Join 

A RIGHT EXCLUDING JOIN returns all rows in Table B that don't match any rows in Table A as shown in the Venn 

diagram. 

 

 

 

PROC SQL ; 

 SELECT * 

  FROM MYDATA.Movies M 

         RIGHT JOIN 

       MYDATA.Actors A 

   ON M.Title = A.Title 

    WHERE M.Title IS NULL ; 

QUIT ; 

 

 

Results for Right Excluding Join: 

 
NOTE: No rows were selected. 
 

 

 

Conclusion 
The objective of this paper was to illustrate the various types of join constructs with the relational model. The SQL 

procedure enables users to join tables of information quickly and easily. Linking one piece of information with 

another piece of information is made possible when at least one column is common to each table. Details about 

the join types were described including the importance of joins, the difference between the various join types 

(inner and outer), the creation of outer (left, right, and full) joins, the importance of the WHERE / ON clause(s) in 

creating and subsetting rows of data, the purpose of creating and using table aliases to minimize the number of 

keystrokes, and the application of set operators. 

